```
In[1]:= SetDirectory["~/KappaLib/"];
<< kappaLib-1.2.m
<< helper.m

Loading KappaLib v1.2

Loading helper.m..
```

- **Define medium**

kappa = C1 ast_g + rho overline (A) otimes A + C2 Id

where g is a Lorentz metric.

- **In suitable coordinates**

```
In[4]:= Metric = k DiagonalMatrix[{-1, 1, 1, 1}];

(* Since hodge operator is conformally invariant,we may assume that k=1. *)
Metric = Metric /. k → 1;

In[6]:= AA = emMatrix["a", 4, Structure → "AntiSymmetric"];

(* rho = scalar density of weight 1 *)
(* C1 = non-zero constant *)
(* C2 = constant *)

In[7]:= kappa = C1 emHodge[Metric] + emBiProduct[rho, AA, AA] + C2 emIdentityKappa[];
```

- **Compute Fresnel polynomial**

```
In[8]:= vars = {x0, x1, x2, x3};
fresnel = FullSimplify[emKappaToFresnel[kappa, vars]];

In[10]:= HH = Table[
    C1 Metric[[i]][[j]] -
     2 rho Sum[AA[[i]][[a]] Metric[[a]][[b]] AA[[b]][[j]], {a, 1, 4}, {b, 1, 4}]
    ,
    {i, 1, 4},
    {j, 1, 4}
   ];

In[11]:= factor1 = vars.Metric.vars;
factor2 = vars.HH.vars;
fresnelExp = - C1^2 factor1 factor2;

In[14]:= FullSimplify[(fresnel - fresnelExp)]

Out[14]= 0
```

Symbolic expression for det(kappa)

```
In[15]:= detKappa = FullSimplify[FullSimplify[emDet[kappa]]]
```

Out[15]= $\left(C1^2 + C2^2\right)^2 \left(C1^2 + 2 \left(-a12^2 - a13^2 - a14^2 + a23^2 + a24^2 + a34^2\right) C1\, rho + C2\, (C2 + 4\, (a14\, a23 - a13\, a24 + a12\, a34)\, rho)\right)$

In[16]:= sub = { EEE → 2 C1 rho $\left(-a12^2 - a13^2 - a14^2 + a23^2 + a24^2 + a34^2\right)$ };

```
In[17]:= detKappaAlt = (C1^2 + C2^2)^2 (C1^2 + C2^2 + EEE + C2 emTrace[emBiProduct[rho, AA, AA]]);
Simplify[(detKappa - detKappaAlt) /. sub]

Out[18]= 0
```

Symbolic expression for det h

In[19]:=
```
FullSimplify[Det[HH]]
```

Out[19]= $-\left(C1^2 + 2\left(-a12^2 - a13^2 - a14^2 + a23^2 + a24^2 + a34^2\right) C1\, rho - 4\, (a14\, a23 - a13\, a24 + a12\, a34)^2\, rho^2\right)^2$

In[20]:=
```
DetMet2 = - (C1^2 + EEE - (1 / 2 emTrace[emBiProduct[rho, AA, AA]]) ^2 )^2;
```

In[21]:=
```
FullSimplify[(Det[HH] - DetMet2) /. sub]
```

Out[21]= 0

Suppose Metric and Metric2 are propertional. Show that this implies that A =0 or rho = 0.

In[22]:=
```
deltaMat = Metric - Const HH;

(* consider only diagonal elements *)
eqs = Table[deltaMat[[i]][[i]], {i, 1, 4}];
```

In[24]:=
```
sub = {};
Case = FullSimplify[eqs //. sub];
show[Case]
```

Out[26]//MatrixForm=

$$\begin{pmatrix} 1 & : & -1 + C1\, Const - 2\left(a12^2 + a13^2 + a14^2\right) Const\, rho \\ 2 & : & 1 - Const\left(C1 + 2\left(-a12^2 + a23^2 + a24^2\right) rho\right) \\ 3 & : & 1 - Const\left(C1 + 2\left(-a13^2 + a23^2 + a34^2\right) rho\right) \\ 4 & : & 1 - Const\left(C1 + 2\left(-a14^2 + a24^2 + a34^2\right) rho\right) \end{pmatrix}$$

- **The second equation shows that Const != 0. Solving for C1 yields**

In[27]:=
```
sub = Append[sub, C1 → 1/Const + 2 (a12^2 + a13^2 + a14^2) rho];
```

In[28]:=
```
Case = FullSimplify[Case //. sub];
show[Case]
```

Out[29]//MatrixForm=

$$\begin{pmatrix} 1 & : & 0 \\ 2 & : & -2\left(a13^2 + a14^2 + a23^2 + a24^2\right) Const\, rho \\ 3 & : & -2\left(a12^2 + a14^2 + a23^2 + a34^2\right) Const\, rho \\ 4 & : & -2\left(a12^2 + a13^2 + a24^2 + a34^2\right) Const\, rho \end{pmatrix}$$

- **Since Const != 0, it follows that rho = 0 or A = 0.**

Extra: Alternative proof that A= 0 using a Gröbner basis

In[30]:=
```
eqs = Union[Flatten[deltaMat]];
gb = GroebnerBasis[eqs, Variables[eqs]]; // Timing
```

Out[31]= {0.029487, Null}

In[32]:= `show[simp[gb]]`

Out[32]//MatrixForm=

$$
\begin{pmatrix}
1 & : & (a14^2 + a23^2)\ rho \\
2 & : & (a13^2 + a24^2)\ rho \\
3 & : & (a12^2 + a34^2)\ rho \\
4 & : & (a13\ a23 + a14\ a24)\ rho \\
5 & : & (a13\ a14 - a23\ a24)\ rho \\
6 & : & (a12\ a24 + a13\ a34)\ rho \\
7 & : & (a12\ a23 - a14\ a34)\ rho \\
8 & : & (a12\ a14 + a23\ a34)\ rho \\
9 & : & (a12\ a13 - a24\ a34)\ rho \\
10 & : & (a14^2 + a23^2)\ (-1 + C1\ Const) \\
11 & : & (a13^2 + a24^2)\ (-1 + C1\ Const) \\
12 & : & (a12^2 + a34^2)\ (-1 + C1\ Const) \\
13 & : & (a13\ a23 + a14\ a24)\ (-1 + C1\ Const) \\
14 & : & (a13\ a14 - a23\ a24)\ (-1 + C1\ Const) \\
15 & : & (a12\ a24 + a13\ a34)\ (-1 + C1\ Const) \\
16 & : & (a12\ a23 - a14\ a34)\ (-1 + C1\ Const) \\
17 & : & (a12\ a14 + a23\ a34)\ (-1 + C1\ Const) \\
18 & : & (a12\ a13 - a24\ a34)\ (-1 + C1\ Const) \\
19 & : & -1 + C1\ Const + 2\ (a23^2 + a24^2 + a34^2)\ Const\ rho \\
20 & : & a12\ (-1 + C1\ Const) + 2\ a34\ (a14\ a23 - a13\ a24 + a12\ a34)\ Const\ rho \\
21 & : & -2\ a14\ a23\ a24\ Const\ rho + a13\ (-1 + C1\ Const + 2\ (a24^2 + a34^2)\ Const\ rho)
\end{pmatrix}
$$

- **The first three equations imply that rho = 0 or A = 0.**